AD-A187 222 COMPUTER PROGRAM TO CALCULATE THE PHYSICAL PROPERTIES 1/1
OF ONE SABOT PETAL(U) ARMY BALLISTIC RESEARCH LAB
ABERDEEN PROVING GROUND MD R A PENNEKAMP JUN 87

UNCLASSIFIED BRL-MR-3598 F/G 19/10 NL





[illegible] REPORT BRL-MR-3598

[illegible] TO CALCULATE [illegible] OF ONE [illegible]

[illegible] A. [illegible]

JUNE 1987

US ARMY BALLISTIC RESEARCH LABORATORY
ABERDEEN PROVING GROUND, MARYLAND

UNCLASSIFIED
SECURITY CLASSIFICATION OF THIS PAGE

AD-A187222

REPORT DOCUMENTATION PAGE

Form Approved
OMB No 0704-0188
Exp Date Jun 30, 1986

1a. REPORT SECURITY CLASSIFICATION: UNCLASSIFIED	1b. RESTRICTIVE MARKINGS
2a. SECURITY CLASSIFICATION AUTHORITY	3. DISTRIBUTION/AVAILABILITY OF REPORT: Approved for public release, distribution unlimited.
2b. DECLASSIFICATION/DOWNGRADING SCHEDULE	
4. PERFORMING ORGANIZATION REPORT NUMBER(S)	5. MONITORING ORGANIZATION REPORT NUMBER(S)
6a. NAME OF PERFORMING ORGANIZATION: U.S. Army Ballistic Research Laboratory; 6b. OFFICE SYMBOL (If applicable): SLCBR-LF	7a. NAME OF MONITORING ORGANIZATION
6c. ADDRESS (City, State, and ZIP Code): Aberdeen Proving Ground, Maryland 21005-5066	7b. ADDRESS (City, State, and ZIP Code)
8a. NAME OF FUNDING/SPONSORING ORGANIZATION: U.S. Army Ballistic Research Laboratory; 8b. OFFICE SYMBOL (If applicable): SLCBR-DD-T	9. PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER
8c. ADDRESS (City, State, and ZIP Code): Aberdeen Proving Ground, Maryland 21005-5066	10. SOURCE OF FUNDING NUMBERS

PROGRAM ELEMENT NO.	PROJECT NO.	TASK NO.	WORK UNIT ACCESSION NO
62618AH80	1L1 62618AH80	NA	

11. TITLE (Include Security Classification)

COMPUTER PROGRAM TO CALCULATE THE PHYSICAL PROPERTIES OF ONE SABOT PETAL

12. PERSONAL AUTHOR(S)

PENNEKAMP, Richard A.

13a. TYPE OF REPORT	13b. TIME COVERED	14. DATE OF REPORT (Year, Month, Day)	15. PAGE COUNT
Memorandum Report	FROM ________ TO ______	April 1987	41

16. SUPPLEMENTARY NOTATION

17. COSATI CODES

FIELD	GROUP	SUB-GROUP
01	01	
20	11	

18. SUBJECT TERMS (Continue on reverse if necessary and identify by block number)

Center of Mass; Mass; Inertia; Tensor

19. ABSTRACT (Continue on reverse if necessary and identify by block number)

A FORTRAN computer program was developed to calculate the following physical properties of one sabot petal: the mass, the center of mass, and the complete inertia tensor about the center of mass. Only uniform density sabot petals are permitted. Any sabot segment angle is allowed up to 360 degrees, which is a complete body of revolution. This report includes the theory, the inertia tensor transformation equations, and listing of the program, and an example problem. Keywords: computer applications, mathematical modeling

(jvs)

20. DISTRIBUTION/AVAILABILITY OF ABSTRACT: ☐ UNCLASSIFIED/UNLIMITED ☐ SAME AS RPT. ☐ DTIC USERS	21. ABSTRACT SECURITY CLASSIFICATION: UNCLASSIFIED
22a. NAME OF RESPONSIBLE INDIVIDUAL: Richard A. Pennekamp	22b. TELEPHONE (Include Area Code): (301) 278-4680; 22c. OFFICE SYMBOL: SLCBR-LF-A

DD FORM 1473, 84 MAR — 83 APR edition may be used until exhausted. All other editions are obsolete.

SECURITY CLASSIFICATION OF THIS PAGE

UNCLASSIFIED

TABLE OF CONTENTS

LIST OF FIGURES

I. INTRODUCTION

A FORTRAN program was written to find the physical properties of one sabot petal. The program was written to provide a simple analytical model to determine these properties. This analytical model is an easy alternative to making actual physical measurements. Though the program was developed for analyzing a sabot petal, it can be used for any homogeneous partial or complete body of revolution. The only restriction is that the body of revolution must be defined by an angle θ_T as shown in Figure 1.

Section II provides the theory for finding the physical properties of one sabot petal. The program takes user-provided contour points of the sabot's RZ plane profile (Figure 1) and uses them to divide the sabot petal into a finite number of partial pieces. The program then finds the physical properties of each partial piece and by using the correct addition formulas, adds all the partial piece results to get the solution for the total sabot petal. Other than the coordinates of the profile, the only information the user provides is the material density of the sabot and the angle of the sabot petal, θ_T.

In Section III, the procedure for transforming the inertia tensor to any Cartesian coordinate is described. All the equations are provided, but the user must be careful of two things. First, the user must not confuse skew lines for parallel lines in the translation transformation equations. Second, the user must follow the "right-handed rule" when determining the sign of the rotation angle in the rotation transformation equations.

Test cases are found in Section IV. The theory is tested on a 25mm aeroballistic model by comparing the results found by physical measurement machines with the answers predicted by the program. The transformation equations are shown to work on a uniform rod.

In Appendix A there is a listing of the program, and in Appendix B an example problem is solved following step-by-step instructions on the execution of the program.

II. THEORY

The sabot petal is divided into partial pieces as required to describe its particular geometry. This is done by describing the profile of the sabot petal in the RZ plane (Figure 1) using contour points. The program uses the contour points to construct the partial pieces and then finds their physical properties. After each partial piece's physical properties are found, the physical properties of the total sabot petal are found by adding the partial piece's properties together using the appropriate addition formulas.

The first physical property found is the volume. The volume for any partial piece "j" of the total sabot petal is:[1]

$$v(j) = \iiint dV \tag{1}$$

If a local cylindrical coordinate system is used (Figure 2), Equation (1) becomes:

$$v(j) = \int_0^{z_t} \int_0^{mz+b} \int_{-\theta_1}^{\theta_1} r \, d\theta dr dz \tag{2}$$

See Figure 3 for integration nomenclature and Figure 2 for the definition of θ_1. The resulting equation for the volume of partial piece "j" is:

$$v(j) = \theta_1\{m^2 z_t^3/3 + z_t^2 mb + b^2 z_t\} \quad . \tag{3}$$

Note that the upper limit of r is restricted to straight line segments. This means that any nonlinear curve in the ZR plane must be approximated by straight line segments. If this is necessary, the approximating error can be minimized by making the number of approximating segments sufficiently large. This is the only approximation used in the program.

The lower limit of r in Equation (2) is always taken as zero. The sabot petal, which is a body with cavities, is handled by the program as follows: The outer surfaces of the sabot define solid bodies of calculable volume; the inner surfaces, which are the boundaries of the cavities, define other solid bodies whose volumes are flagged to be negative. The volume for the total sabot petal is then computed as the sum of the positive and negative volumes. This approach made the programming easier and, more importantly, kept the required input data to a minimum. Figure 4 illustrates the above procedure. This procedure is used for all the physical properties found by this program.

The equation for the volume of the total sabot petal is:

$$V = \sum_{j=1}^{n} v(j) \tag{4}$$

where v(j) is the volume of partial piece "j"
n is the number of partial pieces.

Because only a uniform density is allowed, the total mass of the sabot petal is:

$$M = V\rho \tag{5}$$

where ρ is the density of the material.

With the total mass found, the center of mass can be calculated. The x location for the center of mass of any body is defined as:[2]

$$Xc = \frac{\iiint \rho X \, dV}{\iiint \rho \, dV} \tag{6}$$

Since the density is assumed to be uniform, it can be taken out of the integrals and canceled out. This leaves a denominator equal to the volume of the mass in question. Since the volume of any partial piece "j" has been determined, the only integral that needs to be determined to calculate the x center of mass for any partial piece "j" is:

$$\iiint X \, dV \tag{7}$$

In a local cylindrical coordinate system, Expression (7) becomes

$$\int_0^{z_t} \int_0^{mz+b} \int_{-\theta_1}^{\theta_1} r^2 \cos\theta \, d\theta dr dz \tag{8}$$

Expression (8) is evaluated to be:

$$(2/3)\sin\theta_1\{(1/4)m^3 z_t^{\,4} + m^2 z_t^{\,3} b + (3/2)b^2 m z_t^{\,2} + b^3 z_t\} \tag{9}$$

The x location of the center of mass for partial piece "j" in the global coordinate system is

$$xc(j) = \frac{(2/3)\sin\theta_1\{(1/4)m^3 z_t^{\,4} + m^2 z_t^{\,3} b + (3/2)b^2 m z_t^{\,2} + b^3 z_t\}}{v(j)} \; . \tag{10}$$

The x location of the center of mass for the total sabot petal in the global coordinate system is the sum of the volumes v(j) multiplied by the center of mass xc(j). This value is then divided by the total volume of the sabot petal to give XC, the x location of the center of mass of the total sabot petal in the global coordinate system.[2] This is done in Equation (11).

$$XC = \frac{\sum_{j=1}^{n} v(j)xc(j)}{\sum_{j=1}^{n} v(j)} \tag{11}$$

The definition of the y location of the center of mass is similar to the definition of the x location of the center of mass. The only integration that needs to be performed to determine the y center of mass for any partial piece "j" is:

$$\int_{0}^{z_t} \int_{0}^{mz+b} \int_{-\theta_1}^{\theta_1} r^2 \sin\theta \, d\theta dr dz. \tag{12}$$

However,

$$\int_{-\theta_1}^{\theta_1} \sin\theta \, d\theta = 0 \tag{13}$$

This means the y location of the center of mass of all the "j" partial pieces is 0. This result was expected by the argument of symmetry (Figure 2). By replacing all the "X"'s with "Y"'s in Equation (11) and using the result from Equation (13), the Y location of the center of mass for the total sabot petal, YC, is 0.

The definition of the z location of the center of mass is the same as the definition of the x and y locations of the center of mass. The integral, which is similar to Expressions (8) and (12), is:

$$\int_{0}^{z_t} \int_{0}^{mz+b} \int_{-\theta_1}^{\theta_1} zr \, d\theta dr dz . \tag{14}$$

Expression (14) is evaluated to be:

$$\theta_1\{(1/4)z_t^4m^2 + (2/3)z_t^3mb + (1/2)b^2z_t^2\} \tag{15}$$

The z location for the center of mass of partial piece "j" in the global coordinate system is:

$$zc(j) = ZO + \frac{\theta_1\{(1/4)z_t^4m^2 + (2/3)z_t^3mb + (1/2)b^2z_t^2\}}{v(j)} \tag{16}$$

where ZO is the location of the local coordinate system's origin in the global coordinate system (Figure 2). The z location for the center of mass for the total sabot petal in the global coordinate system is found the same way the x location was found.

$$ZC = \frac{\sum_{j=1}^{n} v(j)zc(j)}{\sum_{j=1}^{n} v(j)} \tag{17}$$

The mass moments of inertia are defined as[2]

$$Ixx = \iiint \rho(y^2 + z^2)\, dV \tag{18}$$

$$Iyy = \iiint \rho(x^2 + z^2)\, dV \tag{19}$$

$$Izz = \iiint \rho(x^2 + y^2)\, dV \tag{20}$$

Equations (18), (19) and (20) for partial piece "j" in its local cylindrical coordinate system are:

$$ixx(j) = \rho \int_0^{z_t} \int_0^{mz+b} \int_{-\theta_1}^{\theta_1} (z^2 r + r^3 \sin^2\theta)\, d\theta\, dr\, dz \tag{21}$$

$$iyy(j) = \rho \int_0^{z_t} \int_0^{mz+b} \int_{-\theta_1}^{\theta_1} (z^2 r + r^3 \cos^2\theta)\, d\theta\, dr\, dz \tag{22}$$

$$izz(j) = \rho \int_0^{z_t} \int_0^{mz+b} \int_{-\theta_1}^{\theta_1} r^3\, d\theta\, dr\, dz \tag{23}$$

Equations (21), (22), and (23) are evaluated to be:

$$ixx(j) = \rho \left\{ \left(\theta_1 - \frac{\sin 2\theta_1}{2}\right) \left(\frac{m^4 z_t^{\,5}}{20} + \frac{bm^3 z_t^{\,4}}{4} + \frac{b^2 m^2 z_t^{\,3}}{2} + \frac{b^3 m z_t^{\,2}}{2} + \frac{b^4 z_t}{4}\right) + \frac{\theta_1 z_t^{\,5} m^2}{5} + \frac{\theta_1 z_t^{\,4} mb}{2} + \frac{\theta_1 z_t^{\,3} b^2}{3} \right\} \tag{24}$$

$$iyy(j) = \rho \left(\theta_1 + \frac{\sin 2\theta_1}{2}\right) \left(\frac{m^4 z_t^{\,5}}{20} + \frac{bm^3 z_t^{\,4}}{4} + \frac{b^2 m^2 z_t^{\,3}}{2} + \frac{b^3 m z_t^{\,2}}{2} + \frac{b^4 z_t}{4}\right) + \frac{\theta_1 z_t^{\,5} m^2}{5} + \frac{\theta_1 z_t^{\,4} mb}{2} + \frac{\theta_1 z_t^{\,3} b^2}{3} \Big\} \tag{25}$$

$$izz(j) = \frac{\rho\theta_1}{2}\left\{\frac{m^4 z_t^5}{5} + m^3 z_t^4 b + 2b^2 m^2 z_t^3 + 2b^3 m z_t^2 + b^4 z_t\right\} . \quad (26)$$

Equations (24), (25), and (26) are partial piece "j"'s mass moments of inertia about its own local coordinate axes. In order to find the mass moments of inertia for the total body, all the partial piece results must be found about axes through the centers of mass of the partial pieces. This is accomplished by the use of the parallel axis theorem:

$$Ixx = Ix'x' + Md^2 \quad (27)$$

where Ix'x' is the moment of inertia about an axis through the center of mass

Ixx is the moment of inertia about an axis that is parallel to an axis through the center of mass

M is the mass of the object

d is the distance between the xx and x'x' axes

The mass moments of inertia of partial piece "j" about axes through its center of mass and parallel to its local system are derived from Equation (27) and the results from Equations (24), (25), and (26). They are:

$$ix'x'(j) = ixx(j) - \rho v(j)\{zc(j) - ZO\}^2 \quad (28)$$

$$iy'y'(j) = iyy(j) - \rho v(j)\{xc(j)^2 + (zc(j) - ZO)^2\} \quad (29)$$

$$iz'z'(j) = izz(j) - \rho v(j) xc(j)^2 . \quad (30)$$

The mass moments of inertia for the total sabot petal about axes through its center of mass and parallel to the global coordinate system are

$$IX'X' = \sum_{j=1}^{n} ix'x'(j) + \rho v(j)(ZC - zc(j))^2 \quad (31)$$

$$IY'Y' = \sum_{j=1}^{n} iy'y'(j) + \rho v(j)[(ZC - zc(j))^2 + (XC - xc(j))^2] \quad (32)$$

$$IZ'Z' = \sum_{j=1}^{n} iz'z'(j) + \rho v(j)(XC - xc(j))^2 \; . \tag{33}$$

Equations (31), (32), and (33) were derived from the parallel axis theorem for mass moments of inertia and Equations (28), (29), and (30).

The mass product of inertia terms are defined as[2]

$$Ixy = \iiint \rho XY \; dV \tag{34}$$

$$Ixz = \iiint \rho XZ \; dV \tag{35}$$

$$Iyz = \iiint \rho YZ \; dV. \tag{36}$$

For partial piece "j" in its local cylindrical coordinate system, Equations (34), (35) and (36) reduce to

$$ixy(j) = \rho \int_0^{z_t} \int_0^{mz+b} \int_{-\theta_1}^{\theta_1} r^3 \cos\theta \sin\theta \; d\theta \; dr \; dz \tag{37}$$

$$ixz(j) = \rho \int_0^{z_t} \int_0^{mz+b} \int_{-\theta_1}^{\theta_1} zr^2 \cos\theta \; d\theta \; dr \; dz \tag{38}$$

$$iyz(j) = \rho \int_0^{z_t} \int_0^{mz+b} \int_{-\theta_1}^{\theta_1} zr^2 \sin\theta \; d\theta \; dr \; dz \; . \tag{39}$$

Equations (37), (38) and (39) are evaluated to be

$$ixy(j) = 0 \tag{40}$$

$$ixz(j) = (2/3)\rho \sin\theta_1 \left[\frac{z_t^{\,5} m^3}{5} + \frac{3}{4} z_t^{\,4} m^2 b + z_t^{\,3} m b^2 + \frac{z_t^{\,2} b^3}{2} \right] \tag{41}$$

$$iyz(j) = 0 \; . \tag{42}$$

Equations (40), (41) and (42) are the mass product of inertia expressions for partial piece "j" about its local coordinate system. Like the mass moments of inertia, the product of inertia terms need to be found about axes through the center of mass of the partial piece "j". This is accomplished by using the parallel axis theorem for products of inertia.[2]

$$Ixy = Ix'y' + M\ Xc\ Yc \tag{43}$$

$$Ixz = Ix'z' + M\ Xc\ Zc \tag{44}$$

$$Iyz = Iy'z' + M\ Yc\ Zc \tag{45}$$

where Ix'y', Ix'z' and Iy'z' are the products of inertia about the center of mass

Ixy, Ixz and Iyz are the products of inertia about a set of axes parallel to the center of mass

Xc, Yc and Zc are the x, y and z locations of the center of mass in an XYZ coordinate system.

Equations (44) and (41) are used to generate the equation for Ixz for partial piece "j" about its own center of mass:

$$ix'z'(j) = ixz(j) - \rho v(j)xc(j)(zc(j) - Z0)\ . \tag{46}$$

Ixz for the total sabot petal about axes through its center of mass and parallel to the global coordinate system, is

$$IX'Z' = \sum_{j=1}^{n} ix'z'(j) + \rho v(j)(xc(j) - XC)(zc(j) - ZC)\ . \tag{47}$$

Equation (47) was arrived at by the use of the parallel axis theorem for products of inertia and Equation (46). Because the y center of mass is zero for every "j" partial piece (Equation (13)) and ixy(j) is zero for every "j" partial piece (Equation (40)), by Equation (43), ix'y'(j) for every "j" partial piece is zero. The total sabot petal's Ix'y', IX'Y', is zero by the same argument. The same argument holds for Iy'z' for the total sabot petal; therefore IY'Z' is zero.

III. INERTIA TENSOR TRANSFORMATIONS

The program calculates the inertia tensor about a set of axes that go through the center of mass of the sabot petal. This section explains how to

transform the inertia tensor properly from this coordinate system to any other Cartesian coordinate system. There are two transformation procedures: translation and rotation. Either procedure can be done independently or they can be done in successive steps. The equations of this section make use of the results found by the program as much as possible for ease of use. Care should be taken if these equations are to be used for anything but this purpose.

Translation of the inertia tensor requires the use of both types of parallel axis theorems used in Section II of this report. Remember to use consistent units and make sure the axes are parallel. The mass moment of inertia terms are translated by Equation (27). With the results found by the program, the general Equation (27) is rewritten as the following equations.

$$I'XX = IX'X' + M[(ZC - Z'0)^2 + (-Y'0)^2] \tag{48}$$

$$I'YY = IY'Y' + M[(XC - X'0)^2 + (ZC - Z'0)^2] \tag{49}$$

$$I'ZZ = IZ'Z' + M[(XC - X'0)^2 + (-Y'0)^2] \tag{50}$$

where X'0, Y'0 and Z'0 are the X, Y and Z locations of the desired coordinate system's origin in the global coordinate system.

For the product of inertia terms, Equations (43), (44) and (45) are rewritten as:

$$I'XY = M(XC - X'0)(-Y'0) \tag{51}$$

$$I'XZ = IX'Z' + M(XC - X'0)(ZC - Z'0) \tag{52}$$

$$I'YZ = M(-Y'0)(ZC - Z'0) \ . \tag{53}$$

Note that all the values in Equations (48-53) are found by the program except X'0, Y'0 and Z'0, which are user-defined.

Only one translation is required to translate the coordinate system of the inertia tensor found by the program to any other coordinate system in space. This is important for the user to know because Equations (48-53) are only valid for the case where the moments and products of inertia are known about axes that go through the center of mass of the body. This means the user cannot take the results from one translation and use Equations (48-53) to make another translation. In general, only one translation is allowed.

Rotation of the coordinate system is allowed and one proper way to do this is presented. The equation to rotate the inertia tensor is:[3]

$$[T'] = [B][T][B]^{-1} \tag{54}$$

where: [T'] is the inertia tensor about the new coordinate system
[T] is the inertia tensor about the old coordinate system
[B] is the transformation rotation tensor
$[B]^{-1}$ is the inverse of the transformation rotation tensor.

Equation (54) allows rotation about a point, but for simplicity only the rotation about any of the coordinate axes will be found. This may require more than one rotation to get to the desired coordinate system, but this is allowed as long as the procedure is followed as directed. Before Equation (54) can be solved, the B tensor must be determined. Figure 5 and Equations (55-63) show how to transform the coordinates of an EFG coordinate system to coordinates in an E'F'G' coordinate system. The transformation is a rotation about the G axis; therefore G' = G.

$$d_1 = f\tan\theta \tag{55}$$

$$L_1 = e + d_1 \tag{56}$$

$$e' = L_1\cos\theta \tag{57}$$

$$e' = e\cos\theta + f\sin\theta \tag{58}$$

$$d_2 = e\tan\theta \tag{59}$$

$$L_2 = f - d_2 \tag{60}$$

$$f' = L_2\cos\theta \tag{61}$$

$$f' = -e\sin\theta + f\cos\theta \tag{62}$$

$$g' = g\ . \tag{63}$$

In matrix form, the rotation transformation from the EFG coordinate system to the E'F'G' coordinate system is:

$$\begin{bmatrix} e' \\ f' \\ g' \end{bmatrix} = \begin{bmatrix} \cos\theta & \sin\theta & 0 \\ -\sin\theta & \cos\theta & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} e \\ f \\ g \end{bmatrix} \tag{64}$$

Therefore

$$[B] = \begin{bmatrix} \cos\theta & \sin\theta & 0 \\ -\sin\theta & \cos\theta & 0 \\ 0 & 0 & 1 \end{bmatrix} . \tag{65}$$

A property of a transformation tensor that rotates one Cartesian coordinate system to another is:[4]

$$[B]^{-1} = [B]^T \tag{66}$$

where the "T" superscript denotes a transposed matrix.

Equations (65) and (66) allow Equation (54) to be written in matrix form as:

$$\begin{bmatrix} IEE' & IEF' & IEG' \\ IFE' & IFF' & IFG' \\ IGE' & IGF' & IGG' \end{bmatrix} = \begin{bmatrix} \cos\theta & \sin\theta & 0 \\ -\sin\theta & \cos\theta & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} IEE & IEF & IEG \\ IFE & IFF & IFG \\ IGE & IGF & IGG \end{bmatrix} \begin{bmatrix} \cos\theta & -\sin\theta & 0 \\ \sin\theta & \cos\theta & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{67}$$

where: θ is the angle of rotation shown in Figure 5.
EFG is the coordinate system shown in Figure 5.

The equations that result from the the solution of Equation (67) are

$$IEE' = IEE\cos^2\theta + IFF\sin^2\theta + IEF\sin 2\theta \tag{68}$$

$$IFF' = IEE\sin^2\theta + IFF\cos^2\theta - IEF\sin 2\theta \tag{69}$$

$$IGG' = IGG \tag{70}$$

$$IEF' = IFE' = .5(IFF - IEE)\sin 2\theta + IEF(\cos^2\theta - \sin^2\theta) \tag{71}$$

$$IEG' = IGE' = IEG\cos\theta + IFG\sin\theta \tag{72}$$

$$IFG' = IGF' = IFG\cos\theta - IEG\sin\theta . \tag{73}$$

Equations (68-73) are generalized equations and to use them in the XYZ coordinate system make the following substitutions. For a rotation about the

X axis	Y axis	Z axis
Let Y = E	Let Z = E	Let X = E
Z = F	X = F	Y = F
X = G	Y = G	Z = G

in Equations (68-73).

The above instructions must be followed to insure a proper transformation from one right-handed Cartesian coordinate system to another. As many rotations as required are allowed using Equations (68-73). If a translation and rotation of the inertia tensor are to be performed, translate by Equations (48-53), then rotate as many times as necessary by Equations (68-73). If a rotation of the axis system is performed, Equations (51) and (53) are in general no longer valid.

IV. TEST CASES

The purpose of the program was to provide an alternative to making actual physical measurements to determine the physical properties of one sabot petal. The test of the program was to compare the physically measured properties with properties found by the program. The sabot petal used for the test is from a 25mm aeroballistic model. The petal is made from aluminum and is one quarter of a complete sabot package. It is shown in Figure 6.

The sabot petal mass, center of mass and mass moments of inertia were found by physical measurement machines located at the Transonic Range facility of Aberdeen Proving Ground. These machines use precision air bearings and have an accuracy of 0.03%. The program determined the properties using 22 contour points. The results for this particular case have a maximum difference of 4.8% when compared with the results from physical measurement equipment. If 203 points are used to more accurately describe the buttress grooves and circular arcs, the maximum difference drops to 2.9%. The comparison of the three cases is shown in Figure 6. This shows a trend towards smaller percentage differences when the number of contour points is increased. If the theory is correct, this trend should occur.

An example of the use of the transformation equations is provided. A uniform rod is divided into six pieces as shown in Figure 7. The inertia properties of each piece are determined by the program. If the transformation equations are correct, the total rod's inertia tensor about its center of mass can be found by transforming all the partial piece results to the location of the center of mass of the total rod and adding them together.

TABLE 1 shows the physical properties calculated by the program for each partial piece. These results are then translated using Equations (48-53). The translated results are shown in TABLE 2. The translated results are then rotated using Equations (68-73) and the results are shown in TABLE 3. With these transformations completed, all the partial pieces' inertia tensors are known about the same set of axes, a set whose origin is the center of mass of the total rod. This allows the results found in TABLE 3 to be added and the sums should equal the inertia terms for the total rod.

TABLE 1. Program-determined Properties.

	1	2	3	4	5	6
MASS	.3173	.0529	.4760	.3173	1.481	.5288
XC	.4244	.6591	.6002	.6366	.3514	.4919
ZC	1.000	1.000	3.000	3.000	4.000	2.000
IX'X'	.1851	.0182	1.471	.9656	8.318	.8121
IY'Y'	.1279	.0205	1.451	.9682	8.034	.7346
IZ'Z'	.1015	.0035	.0665	.0301	.5575	.1364
IX'Z' $\times 10^7$	-.149	-.037	-.596	.000	-2.38	0.000

TABLE 2. Tensor Translations.

	1	2	3	4	5	6
Ixx	5.257	.866	3.373	2.234	9.799	5.573
Iyy	5.257	.892	3.526	2.365	9.697	5.624
Izz	.158	.026	.238	.159	.740	.264
Ixy	0.000	0.000	0.000	0.000	0.000	0.000
Ixz	-.538	-.139	-.571	-.403	.520	.781
Iyz	0.000	0.000	0.000	0.000	0.000	0.000

TABLE 3. Tensor Rotations.

	1	2	3	4	5	6	Sum 1-6
Ixx	5.257	.890	3.384	2.332	9.792	5.576	27.231
Iyy	5.257	.868	3.516	2.267	9.704	5.621	27.233
Izz	.158	.026	.238	.159	.740	.264	1.585
Ixy	0.000	.007	.038	-.057	.026	-.013	.001
Ixz	-.538	.036	.552	.202	.502	-.754	0.000
Iyz	0.000	-.134	-.148	.349	.135	-.202	0.000

The inertia terms for a homogeneous right circular cylinder are found by the following equations:[2]

$$Ixx = Iyy = (1/12)M(3r^2 + h^2) \tag{74}$$

$$Izz = (1/2)Mr^2 \tag{75}$$

$$Ixy = Ixz = Iyz = 0 \tag{76}$$

where M is the mass of the rod, r is the radius of the circular cross section and h is the height of the cylinder.

The inertia terms for the total rod are

$$Ix'x' = Iy'y' = 27.235 \text{ lb-in}^2 \tag{77}$$

$$Iz'z' = 1.587 \text{ lb-in}^2 \tag{78}$$

$$Ix'y' = Ix'z' = Iy'z' = 0 \tag{79}$$

If the transformation equations are correct, the expected results found by Equations (77-79) should equal the results found in the sum row in TABLE 3. It has been shown that for this particular problem, the transformation equations work. The small differences in the expected results are due to round-off error.

V. CONCLUSIONS

Inertia properties are sometimes important to know. Shapes like sabot petals usually make it difficult to calculate inertia terms analytically. The alternative to an analytical calculation is to make actual physical measurements. Access to physical measurement equipment is sometimes difficult and getting the necessary information can take some time. The intention of this program is to provide this information quickly and accurately.

The inertia properties of an object are defined about coordinate axes. Because the inertia information may be needed about axes other than the defined axes, transformation equations were derived. The report includes all the equations needed to translate and rotate the inertia tensor to the required coordinate system. The equations were shown to work in a selected example. The equations should make the process of coordinate transformation easier for the user.



Figure 1. Global coordinate system.



Figure 2. Local coordinate system.



Figure 3. Integration nomenclature.



Figure 4. Lower limit of r explanation.



Figure 5. Rotation transformation relationships.



Figure 6. Machine vs program test case.



	1	2	3	4	5	6
θ	180	30	90	60	210	150
l	2	2	6	6	8	4
ϕ	0	105	165	240	15	195
Z'	5	5	5	5	3	-1

NOTES

1. WHEN THE 6 PIECES ARE ASSEMBLED, THEY MAKE A 10" LONG RIGHT CIRCULAR CYLINDER WITH A 1" RADIUS
2. MATERIAL DENSITY .101 LBS./IN.^{3}
3. UNITS- INCHES AND LBS.(FORCE)
4. Z'- LOCATION OF THE TOTAL ROD'S Z CENTER OF MASS IN EACH PIECE'S GLOBAL COORDINATE SYSTEM
5. X & Y LOCATION OF THE CENTER OF MASS OF THE ROD OCCUR AT (0,0) IN EVERY PIECE'S GLOBAL SYSTEM
6. ϕ - ANGLE BETWEEN THE ROD'S X AXIS AND EACH PIECE'S X AXIS
7. X̄ & Ȳ ARE THE TOTAL RODS AXES

Figure 7. Inertia tensor transformation example.

LIST OF REFERENCES

1. R. Ellis and D. Gullick, "Calculus with Analytic Geometry," Chapter 14, p-p. 881, Harcourt Brace Jovanovich, Inc., NY, 1978.

2. I. H. Shames, "Engineering Mechanics," Chapters 8-9, pp. 302-350, Prentice-Hall Inc., Englewood Cliffs, NJ, 1980.

3. G. E. Mase, "Theory and Problems of Continuum Mechanics," Chapter 1, pp. 15, McGraw-Hill Book Company, NY, 1970.

4. Y. C. Fung, "A First Course in Continuum Mechanics," Chapter 2, pp. 48, Prentice-Hall Inc., Englewood Cliffs, NJ, 1977.

LIST OF SYMBOLS

b	local variable; r intercept; used with m to define the upper bound of integration on r
d	distance
B	rotation transformation tensor
dr	local r differential length
dV	differential volume
dz	local z differential length
e	Cartesian coordinate used to define the rotation transformation tensor
e'	Cartesian coordinate that results after a rotation about the G axis
E	Cartesian axis used to define the rotation transformation tensor
E'	Cartesian axis that results after a rotation about G axis
f	Cartesian coordinate used to define the rotation transformation tensor
f'	Cartesian coordinate that results after a rotation about the G axis
F	Cartesian axis used to define the rotation transformation tensor
F'	Cartesian axis that results after a rotation about G axis
G	Cartesian axis used to define the rotation transformation tensor
IEE	mass moment of inertia about the E axis
IEE'	mass moment of inertia about the E' axis
IEF	mass product of inertia about the E and F axes
IEF'	mass product of inertia about the E' and F' axes
IEG	mass product of inertia about the E and G axes
IEG'	mass product of inertia about the E' and G' axes
IFF	mass moment of inertia about the F axis
IFF'	mass moment of inertia about the F' axis
IFG	mass product of inertia about the F and G axes

LIST OF SYMBOLS (Continued)

IFG' mass product of inertia about the F' and G' axes

IGG mass moment of inertia about the G axis

IGG' mass moment of inertia about the G' axis

ixx(j) partial piece "j"'s mass moment of inertia about its own local x axis

ix'x'(j) partial piece "j"'s mass moment of inertia about an axis through its center of mass and parallel to its own local x axis

Ixx a body's mass moment of inertia about the X axis in an XYZ coordinate system

Ix'x' a body's mass moment of inertia about an axis through its center of mass and parallel to the X axis in an XYZ coordinate system

IX'X' the total sabot's mass moment of inertia about an axis through its center of mass and parallel to the X global axis

I'XX mass moment of inertia of the total sabot about a user-defined X axis

ixy(j) partial piece "j"'s mass product of inertia about its own local x and y axes

ix'y'(j) partial piece "j"'s mass product of inertia about axes through its center of mass and parallel to its own local x and y axes

Ixy a body's mass product of inertia about the X and Y axes in an XYZ coordinate system

Ix'y' a body's mass product of inertia about axes through its center of mass and parallel to the X and Y axes in an XYZ coordinate system

IX'Y' the sabot's total mass product of inertia about axes through its center of mass and parallel to the global X and Y axes

I'XY mass product of inertia of the total sabot about user-defined X and Y axes

ixz(j) partial piece "j"'s mass product of inertia about its own local x and z axes

ix'z'(j) partial piece "j"'s mass product of inertia about axes through its center of mass and parallel to its own local x and z axes

Ixz a body's mass product of inertia about the X and Z axes in an XYZ coordinate system

LIST OF SYMBOLS (Continued)

Ix'z'	a body's mass product of inertia about axes through its center of mass and parallel to the X and Z axes in an XYZ coordinate system
IX'Z'	the sabot's total mass product of inertia about axes through its center of mass and parallel to the global X and Z axes
I'XZ	mass product of inertia of the total sabot about user defined X and Z axes
iyy(j)	partial piece "j"'s mass moment of inertia about its own local y axis
iy'y'(j)	partial piece "j"'s mass moment of inertia about an axis through its center of mass and parallel to its own local y axis
Iyy	a body's mass moment of inertia about the Y axis in an XYZ coordinate system
Iy'y'	a body's mass moment of inertia about an axis through its center of mass and parallel to the Y axis in an XYZ coordinate system
IY'Y'	the total sabot's mass moment of inertia about an axis through its center of mass and parallel to the Y global axis
I'YY	mass moment of inertia of the total sabot about a user-defined Y axis
iyz(j)	partial piece "j"'s mass product of inertia about its own local y and z axes
iy'z'(j)	partial piece "j"'s mass product of inertia about axes through its center of mass and parallel to its own local y and z axes
Iyz	a body's mass product of inertia about the Y and Z axes in an XYZ coordinate system
Iy'z'	a body's mass product of inertia about axes through its center of mass and parallel to the Y and Z axes in an XYZ coordinate system
IY'Z'	the sabot's total mass product of inertia about axes through its center of mass and parallel to the global Y and Z axes
I'YZ	mass product of inertia of the total sabot about user-defined Y and Z axes
izz(j)	partial piece "j"'s mass moment of inertia about its own local z axis
iz'z'(j)	partial piece "j"'s mass moment of inertia about an axis through its center of mass and parallel to its own local z axis

LIST OF SYMBOLS (Continued)

Izz	a body's mass moment of inertia about the Z axis in an XYZ coordinate system
Iz'z'	a body's mass moment of inertia about an axis through its center of mass and parallel to the Z axis in an XYZ coordinate system
IZ'Z'	the total sabot's mass moment of inertia about an axis through its center of mass and parallel to the Z global axis
I'ZZ	mass moment of inertia of the total sabot about a user defined Z axis
j	used as an index to denote any partial piece used to describe the total body
m	local variable; dr/dz; used with b to define the upper bound of integration on r
M	mass
n	number of partial pieces
r	local cylindrical coordinate
T	inertia tensor
T'	inertia tensor after a coordinate transformation
v(j)	partial piece "j"'s volume
V	the sabot's total volume
x	local Cartesian coordinate
X	Cartesian coordinate
X'O	X location of a user defined system's origin in the global coordinate system
xc(j)	partial piece "j"'s X value of the center of mass in the global coordinate system
Xc	a body's X value of center of mass
XC	the total sabot's X value of the center of mass in the global coordinate system
y	local Cartesian coordinate
Y	Cartesian coordinate

LIST OF SYMBOLS (Continued)

Y'0	Y location of a user defined system's origin in the global coordinate system
yc(j)	partial piece "j"'s Y value of the center of mass in the global coordinate system
Yc	a body's Y value of center of mass
YC	the total sabot's Y value of the center of mass in the global coordinate system
z	local Cartesian and cylindrical coordinate
Z	Cartesian and cyclindrical coordinate
zc(j)	partial piece "j"'s Z value of the center of mass in the global coordinate system
Z0	Z location of the partial piece "j"'s origin in the global coordinate system
Z'0	Z location of a user defined system's origin in the global coordinate system
Zc	a body's Z value of center of mass
ZC	the total sabot's Z value of the center of mass in the global coordinate system
z_t	local variable; z length of partial piece "j" and limit of integration on z

Greek Symbols

$d\theta$	angular differential
θ	cylindrical angular coordinate
θ_1	limit of integration in local coordinate system; see Figure 2
θ_T	total angular measurement on one sabot petal; see Figure 1; note that $2\theta_1 = \theta_T$
ρ	density
Σ	summation sign

APPENDIX A. LISTING OF THE PROGRAM

General Notes

1. Program is written in FORTRAN 77

2. Arrays Z and R hold the body contour points. The dimension of the arrays is set at 100. This allows for 100 contours to describe the body. This can be increased if necessary.

3. Array D(j,k) holds all the information about each partial piece.

j=number of the partial piece

k=1 holds the volume of partial piece "j"
k=2 holds the XC in the global system of partial piece "j"
k=3 holds the ZC in the global system of partial piece "j"
k=4 holds the IX'X' of partial piece "j"
k=5 holds the IY'Y' of partial piece "j"
k=6 holds the IZ'Z' of partial piece "j"
k=7 holds the IX'Z' of partial piece "j".

The dimension of the D array is set at (100,7). If more than 100 partial pieces are needed to describe the body, increase "j" only. Leave "k" set at 7.

4. The computer for which the FORTRAN program was written uses I/O unit "5" to read information from the screen and "6" to write information to the screen. If this is not the case on the system where this program is going to be placed, the correct read and write numbers will have to be exchanged for "5" and "6" in the program listing.

5. The important variable names in the program are described.

All "TOT" variables like "VOLTOT" and "IXZTOT" are the values for the total body.

All "PAR" variables like "VOLPAR" and "XCPAR" are partial values for the total body.

NUMREG = number of partial pieces.

NUMPTS = number of contour points to describe the body.

THDEG = θ_T in degrees. See Figure 1.

TH = θ_1 in radians. See Figure 2.

RHO = density of the material. Units are defined by user's entries.

B = r intercept of the top boundary of a partial piece in its own coordinate system. Units are defined by user's entries. It is b in Figure 3.

M = slope (dr/dz) of the top boundary of a partial piece in its own coordinate system. It is m in Figure 3.

Z = length of a partial piece. Units are defined by user's entries. It is z_t in Figure 3.

"T" variables like T1 or T5 are used for clarity. They represent terms in the large physical property equations.

```
*
* PROGRAM TO COMPUTE THE PHYSICAL PROPERTIES OF ONE SABOT PETAL
*
* DIMENSION AND DATA ENTRY
*
      DIMENSION D(100,7),R(100),Z(100)
      REAL IXXTOT,IYYTOT,IZZTOT,IXZTOT
      CHARACTER FNAME*45
      PI=3.1415927
      WRITE(6,*)'ENTER FILE NAME'
      READ(5,101)FNAME
      OPEN(10,FILE=FNAME,STATUS='OLD')
      I=1
4     READ(10,*)Z(I),R(I)
      IF(I.EQ.1)GOTO8
      IF((Z(I).EQ.Z(1)).AND.(R(I).EQ.R(1)))GOTO9
8     I=I+1
      GOTO4
9     READ(10,*)RHO,THDEG
      CLOSE(10)
      TH=THDEG*PI/360.
3     NUMPTS=I-1
      NUMREG=0
*
* FINDS THE PROPERTIES OF ALL THE PARTIAL PIECES AND STORES THEM IN
* THE "D" ARRAY
*
      DO 5 K=1,NUMPTS
      IF(Z(K).EQ.Z(K+1))GOTO5
      NUMREG=NUMREG+1
      L=NUMREG
      CALL VOLCEN(R(K+1),R(K),Z(K+1),Z(K),D(L,1),D(L,2),D(L,3),TH)
      IF (D(L,1).EQ.0.)GOTO5
      CALL RMXANY(R(K+1),R(K),Z(K+1),Z(K),D(L,1),D(L,2),D(L,3),
     :D(L,4),D(L,5),TH,RHO)
      CALL RMOIZZ(R(K+1),R(K),Z(K+1),Z(K),D(L,1),D(L,2),D(L,6),TH,RHO)
      CALL RMOIXZ(R(K+1),R(K),Z(K+1),Z(K),D(L,1),D(L,2),D(L,3),
     :D(L,7),TH,RHO)
5     CONTINUE
*
* FINDS THE VOLUME,X AND Z LOCATION OF THE CENTER OF MASS FOR THE SABOT PETAL
*
      VOLPAR=0.
      XCPAR=0.
```

```
      ZCPAR=0.
      DO 6 I=1,NUMREG
      VOLPAR=VOLPAR+D(I,1)
      XCPAR=XCPAR+D(I,1)*D(I,2)
6     ZCPAR=ZCPAR+D(I,1)*D(I,3)
* (EQ.4)
      VOLTOT=VOLPAR
* (EQ.11)
      XCTOT=XCPAR/VOLTOT
* (EQ.17)
      ZCTOT=ZCPAR/VOLTOT
*
* CALCULATES IX'X',IY'Y',IZ'Z', AND IX'Z' FOR THE SABOT PETAL
*
      DO 7 I=1,NUMREG
* (EQ.31)
      IXXTOT=IXXTOT+D(I,4)+(ZCTOT-D(I,3))*(ZCTOT-D(I,3))*RHO*D(I,1)
* (EQ.32)
      IYYTOT=IYYTOT+D(I,5)+((XCTOT-D(I,2))*(XCTOT-D(I,2))+
     :                  (ZCTOT-D(I,3))*(ZCTOT-D(I,3)))*RHO*D(I,1)
* (EQ.33)
      IZZTOT=IZZTOT+D(I,6)+(XCTOT-D(I,2))*(XCTOT-D(I,2))*RHO*D(I,1)
* (EQ.47)
7     IXZTOT=IXZTOT+D(I,7)+(D(I,2)-XCTOT)*(D(I,3)-ZCTOT)*RHO*D(I,1)
*
* DATA OUTPUT AND FORMAT STATEMENTS
*
      WRITE(6,102)FNAME
      WRITE(6,103)RHO
      WRITE(6,104)THDEG
      WRITE(6,105)VOLTOT*RHO
      WRITE(6,106)XCTOT
      WRITE(6,107)
      WRITE(6,108)ZCTOT
      WRITE(6,109)IXXTOT
      WRITE(6,110)IYYTOT
      WRITE(6,111)IZZTOT
      WRITE(6,112)
      WRITE(6,113)IXZTOT
      WRITE(6,114)
101   FORMAT(A45)
102   FORMAT(' ','FILE NAME: ',A45)
103   FORMAT(' ','DENSITY: ',G11.4E2)
104   FORMAT(' ','THETA T: ',F6.2,' DEGREES',/)
105   FORMAT(' ','MASS: ',G12.6E2,/)
106   FORMAT(' ','XC: ',G12.6E2)
107   FORMAT(' ','YC: 0.')
108   FORMAT(' ','ZC: ',G12.6E2,/)
109   FORMAT(' ','IX''X'': ',G12.6E2)
110   FORMAT(' ','IY''Y'': ',G12.6E2)
111   FORMAT(' ','IZ''Z'': ',G12.6E2,/)
112   FORMAT(' ','IX''Y'': 0.')
113   FORMAT(' ','IX''Z'': ',G12.6E2)
114   FORMAT(' ','IY''Z'': 0.')
```

```
      END
*
* VOLCEN CALCULATES THE VOLUME AND CENTER OF MASS FOR ANY PARTIAL PIECE
*
      SUBROUTINE VOLCEN(R2,R1,Z2,Z1,VOL,XC,ZC,TH)
      REAL M
      B=R1
      M=(R2-R1)/(Z2-Z1)
      Z=Z2-Z1
      T1=M*M*Z*Z*Z/3.
      T2=B*M*Z*Z
      T3=B*B*Z
* (EQ.3)
      VOL=TH*(T1+T2+T3)
      IF(VOL.EQ.0)RETURN
      T4=M*M*M*Z*Z*Z*Z/4.
      T5=B*M*M*Z*Z*Z
      T6=B*B*Z*Z*M*3./2.
      T7=B*B*B*Z
* (EQ.10)
      XC=2./3.*SIN(TH)*(T4+T5+T6+T7)/VOL
      T8=M*M*Z*Z*Z*Z/4.
      T9=B*M*Z*Z*Z*2./3.
      T10=B*B*Z*Z/2.
* (EQ.16)
      ZC=Z1+(TH*(T8+T9+T10))/VOL
      RETURN
      END
*
* RMXANY CALCULATES IX'X' AND IY'Y' FOR ANY PARTIAL PIECE
*
      SUBROUTINE RMXANY(R2,R1,Z2,Z1,VOL,XC,ZC,IXX,IYY,TH,RHO)
      REAL M,IXX,IYY
      B=R1
      M=(R2-R1)/(Z2-Z1)
      Z=Z2-Z1
      T1=M*M*M*M*Z*Z*Z*Z*Z/20.
      T2=B*M*M*M*Z*Z*Z*Z/4.
      T3=B*B*M*M*Z*Z*Z/2.
      T4=B*B*B*M*Z*Z/2.
      T5=B*B*B*B*Z/4.
      T6=Z*Z*Z*Z*Z*M*M/5.
      T7=Z*Z*Z*Z*M*B/2.
      T8=Z*Z*Z*B*B/3.
      X=TH-.5*SIN(2*TH)
* (EQ.24)
      IXX=RHO*(X*(T1+T2+T3+T4+T5)+TH*(T6+T7+T8))
* (EQ.28)
          IXX=IXX-(ZC-Z1)*(ZC-Z1)*VOL*RHO
      X=TH+.5*SIN(2*TH)
* (EQ.25)
      IYY=RHO*(X*(T1+T2+T3+T4+T5)+TH*(T6+T7+T8))
* (EQ.29)
          IYY=IYY-(XC*XC+(ZC-Z1)*(ZC-Z1))*VOL*RHO
```

```
      RETURN
      END
*
* RMOIZZ CALCULATES IZ'Z' FOR ANY PARTIAL PIECE
*
      SUBROUTINE RMOIZZ(R2,R1,Z2,Z1,VOL,XC,IZZ,TH,RHO)
      REAL M,IZZ
      B=R1
      M=(R2-R1)/(Z2-Z1)
      Z=Z2-Z1
      T1=M*M*M*M*Z*Z*Z*Z*Z/5.
      T2=B*M*M*M*Z*Z*Z*Z
      T3=2.*B*B*M*M*Z*Z*Z
      T4=B*B*B*M*Z*Z*2.
      T5=B*B*B*B*Z
* (EQ.26)
      IZZ=RHO*TH*.5*(T1+T2+T3+T4+T5)
* (EQ.30)
      IZZ=IZZ-VOL*RHO*XC*XC
      RETURN
      END
*
* CALCULATES IX'Z' FOR ANY PARTIAL PIECE
*
      SUBROUTINE RMOIXZ(R2,R1,Z2,Z1,VOL,XC,ZC,IXZ,TH,RHO)
      REAL M,IXZ
      B=R1
      M=(R2-R1)/(Z2-Z1)
      Z=Z2-Z1
      T1=Z*Z*Z*Z*Z*M*M*M/5.
      T2=Z*Z*Z*Z*M*M*B*.75
      T3=Z*Z*Z*M*B*B
      T4=Z*Z*B*B*B/2.
* (EQ.41)
      IXZ=RHO*2./3.*SIN(TH)*(T1+T2+T3+T4)
* (EQ.46)
      IXZ=IXZ-RHO*VOL*XC*(ZC-Z1)
      RETURN
      END
```

APPENDIX B. EXAMPLE PROBLEM

The physical properties of a sabot with the a ZR profile shown in Figure B-1 are to be determined. The sabot petal is made of aluminum and is one-third of a complete sabot package.

STEP 1 FIND THE COORDINATES OF THE CONTOUR POINTS IN THE ZR PLANE. SEE FIGURE B-1.

STEP 2 CREATE A DATA FILE OF THE FORM OF FIGURE B-2.

This file contains the coordinates found in STEP 1. The data file entries must follow the contour of the body in a clockwise manner until the first point is reached. The first point is then entered again. This is necessary because this signals to the computer that the profile is completely entered. The choice of the first point is arbitrary.

The last record of the data file has two values. The first value is the density and the second value is the angle θ_T. See Figure 1. For the example problem, $\theta_T = 120$. Aluminum, the material of the example problem, has a density of .00314 slug/inch.[3] Please note that the length units of density are the same as the length units used to describe the ZR profile.

STEP 3 EXECUTE THE PROGRAM

The only entry required is the name of the file created in STEP 2.

STEP 4 SOLUTION

The solution is written to the screen of the terminal. If a terminal is not used, the program will have to be modified by the user. The example problem's solution is shown in Figure B-3. The units are consistent with the units entered in STEP 2. For the example problem the mass is in slugs, the X and Z locations of the center of mass are in inches from the origin of the global coordinate system (Figure 1) and the moments and product of inertia are in slugs inches.[2] Remember the moments of inertia are determined about axes through the center of mass and parallel to the global coordinate system.



	Z	R
1	0.	1.5
2	.5	1.5
3	1.	.75
4	2.	.9
5	2.	1.5
6	2.5	1.5
7	2.5	.9
8	3.5	.5
9	1.	.5
10	.5	1.25
11	0.	1.25

UNITS - INCHES

Figure B-1. Example problem contour.

0	1.5
.5	1.5
1	.75
2	.9
2	1.5
2.5	1.5
2.5	.9
3.5	.5
1	.5
.5	1.25
0	1.25
0	1.5
.00314	120.

Figure B-2. Example problem data file.

FILE NAME: EPDF
DENSITY: 0.3140E-02
THETA T: 120.00 DEGREES

MASS: 0.749573E-02

XC: 0.826570
YC: 0.
ZC: 1.71011

IX'X': 0.791766E-02
IY'Y': 0.620526E-02
IZ'Z': 0.312264E-02

IX'Y': 0.
IX'Z': -.657281E-03
IY'Z': 0.

Figure B-3. Example problem solution.

DISTRIBUTION LIST

No. of Copies	Organization
12	Administrator Defense Technical Info Center ATTN: DTIC-FDAC Cameron Station, Bldg. 5 Alexandria, VA 22304-6145
1	HQDA DAMA-ART-M Washington, DC 20310
1	Commander US Army Materiel Command ATTN: AMCDRA-ST 5001 Eisenhower Avenue Alexandria, VA 22333-0001
1	Commander US Army Armament Research, Development and Engineering Center ATTN: SMCAR-MSI Dover, NJ 07801-5001
1	Commander US Army Armament Research Development and Engineering Center ATTN: SMCAR-TDC Dover, NJ 07801-5001
2	Commander US Army Armament Research Development and Engineering Center ATTN: SMCAR-AET Mr. A. Loeb Mr. D. Mertz Dover, NJ 07801-5001
1	Commander US Army Armament Research Development and Engineering Center ATTN: SMCAR-LCU-CT Mr. Lee Whitmore Dover, NJ 07801-5001
1	Commander US AMCCOM ARDEC CCAC Benet Weapons Laboratory ATTN: SMCAR-CCB-TL Watervliet, NY 12189-4050
1	Commander US Army Armament, Munitions and Chemical Command ATTN: AMSMC-IMP-L Rock Island, IL 61299-7300
1	Commander US Army Aviation Systems Command ATTN: AMSAV-ES 4300 Goodfellow Blvd St. Louis, MO 63120-1798
1	Director US Army Aviation Research and Technology Activity Ames Research Center Moffett Field, CA 94035-1099
10	C.I.A. OIR/DB/Standard GE47 HQ Washington, DC 20505
1	Commander US Army Communications - Electronics Command ATTN: AMSEL-ED Fort Monmouth, NJ 07703-5301
1	Commander CECOM R&D Technical Library ATTN: AMSEL-IM-L (Reports Section) B.2700 Fort Monmouth, NJ 07703-5000
1	Commander US Army Missile Command Research, Development, and Engineering Center ATTN: AMSMI-RD Redstone Arsenal, AL 35898-5230

DISTRIBUTION LIST

No. of Copies	Organization
1	Director US Army Missile and Space Intelligence Center ATTN: AIAMS-YDL Redstone Arsenal, AL 35898-5500
1	Commander US Army Tank Automotive Command ATTN: AMSTA-TSL Warren, MI 48397-5000
1	Director US Army TRADOC Analysis Center ATTN: ATOR-TSL White Sands Missile Range, NM 88002-5502
1	Commandant US Army Infantry School ATTN: ATSH-CD-CS-OR Fort Benning, GA 31905-5400
1	Commander US Army Development and Employment Agency ATTN: MODE-ORO Fort Lewis, WA 98433-5000
1	AFWL/SUL Kirtland AFB, NM 87117
1	Air Force Armament Laboratory ATTN: AFATL/DLODL (Tech Info Center) Eglin AFB, FL 32542-5000

No. of Copies	Organization
	Aberdeen Proving Ground
	Dir, USAMSAA ATTN: AMXSY-D AMXSY-MP, H. Cohen
	Cdr, USATECOM ATTN: AMSTE-SI-F
	Cdr, CRDC, AMCCOM ATTN: SMCCR-RSP-A SMCCR-MU SMCCR-SPS-IL

USER EVALUATION SHEET/CHANGE OF ADDRESS

This Laboratory undertakes a continuing effort to improve the quality of the reports it publishes. Your comments/answers to the items/questions below will aid us in our efforts.

1. BRL Report Number____________________Date of Report__________

2. Date Report Received______________________________

3. Does this report satisfy a need? (Comment on purpose, related project, or other area of interest for which the report will be used.)__________

4. How specifically, is the report being used? (Information source, design data, procedure, source of ideas, etc.)__________

5. Has the information in this report led to any quantitative savings as far as man-hours or dollars saved, operating costs avoided or efficiencies achieved, etc? If so, please elaborate.__________

6. General Comments. What do you think should be changed to improve future reports? (Indicate changes to organization, technical content, format, etc.)

CURRENT ADDRESS

Name

Organization

Address

City, State, Zip

7. If indicating a Change of Address or Address Correction, please provide the New or Correct Address in Block 6 above and the Old or Incorrect address below.

OLD ADDRESS

Name

Organization

Address

City, State, Zip

(Remove this sheet, fold as indicated, staple or tape closed, and mail.)

FOLD HERE

Director
US Army Ballistic Research Laboratory
ATTN: DRXBR-OD-ST
Aberdeen Proving Ground, MD 21005-5066

NO POSTAGE
NECESSARY
IF MAILED
IN THE
UNITED STATES

OFFICIAL BUSINESS
PENALTY FOR PRIVATE USE, $300

BUSINESS REPLY MAIL
FIRST CLASS PERMIT NO 12062 WASHINGTON, DC

POSTAGE WILL BE PAID BY DEPARTMENT OF THE ARMY

Director
US Army Ballistic Research Laboratory
ATTN: DRXBR-OD-ST
Aberdeen Proving Ground, MD 21005-9989

FOLD HERE

END
DATE
FILMED
FEB.
1988